!
! SYS_SIZE is the number of clicks (16 bytes) to be loaded.
! 0x3000 is 0x30000 bytes = 196kB, more than enough for current
! versions of linux
! SYS_SIZE 是要加载的节数（16字节为一节）。0x3000工位0x30000字节=196kB，
! 对于当前版本空间已经足够
SYSSIZE = 0x3000 ! 只编译链接活system模块的大小，这里给出了一个最大值
!
!	bootsect.s		(C) 1991 Linus Torvalds
!
! bootsect.s is loaded at 0x7c00 by the bios-startup routines, and moves
! iself out of the way to address 0x90000, and jumps there.
! bootsect.s 被 bios-启动子程序加载至0x7c00（31K）处，并将自己移到地址
! 0x90000（576k）处，并跳转至那里
! 
! It then loads 'setup' directly after itself (0x90200), and the system
! at 0x10000, using BIOS interrupts. 
! 然后使用BIOS中断将'setup'直接加载到自己的后面（0x90200）（576.5k），并将
! system加载到地址0x10000处
! 
! NOTE! currently system is at most 8*65536 bytes long. This should be no
! problem, even in the future. I want to keep it simple. This 512 kB
! kernel size should be enough, especially as this doesn't contain the
! buffer cache as in minix
! 
! The loader has been made as simple as possible, and continuos
! read errors will result in a unbreakable loop. Reboot by hand. It
! loads pretty fast by getting whole sectors at a time whenever possible.

.globl begtext, begdata, begbss, endtext, enddata, endbss ! 定义了6个全局标识符
.text  ! 文本段
begtext:
.data  ! 数据段
begdata:
.bss   ! 未初始化数据段（Block Started by Symbol）
begbss:
.text  ! 文本段

! setup程序的扇区数（setup-sectors）值
SETUPLEN = 4				! nr of setup-sectors
! bootsect的原始地址（是段地址，以下同）
BOOTSEG  = 0x07c0			! original address of boot-sector
! 将bootsect移到这里 -- 避开
INITSEG  = 0x9000			! we move boot here - out of the way
! setup程序从这里开始
SETUPSEG = 0x9020			! setup starts here
! system加载到地址0x10000（64KB）处
SYSSEG   = 0x1000			! system loaded at 0x10000 (65536).
停止加载的段地址
ENDSEG   = SYSSEG + SYSSIZE		! where to stop loading

! ROOT_DEV:	0x000 - same type of floppy as boot. 根文件系统设备使用与引导时同样的软驱设备
!		0x301 - first partition on first drive etc 根文件系统设备在第一个硬盘的第一个分区上，等等
! 指定根文件系统设备时第2个硬盘的第一个分区。
! 这是Linux老式的硬盘命名方法，具体值的含义如下：
! 设备号=主设备号*256 + 次设备号
! 也就是 dev_no = (major<<8) + minor
! 主设备号：1—内存，2-磁盘，3-硬盘，4-ttyx，5-tty，6-并行口，7-非命名管道
ROOT_DEV = 0x306

! 告知连接程序，程序从start标号开始执行
entry start
! 这一段的作用式将自身（bootsect）从目前段位置0x07c0（31K）移动到0x9000（576k）
! 处，共256字（512字节），然后跳转到移动后代码的go标号处，也即本程序的下一语句处
start:
	mov	ax,#BOOTSEG ! 将ds段寄存器置为0x07c0
	mov	ds,ax
	mov	ax,#INITSEG ! 将es段寄存器置为0x9000
	mov	es,ax
	mov	cx,#256     ! 移动计数值=256字
	sub	si,si       ! 源地址 ds:si = 0x07c0:0x0000
	sub	di,di       ! 目的地址 es:di = 0x9000:0x0000
	rep             ! 重复执行，直到cx为0
	movw            ! 移动1个字
	jmpi	go,INITSEG ! 间接跳转，这里的INITSEG指出跳转到的地址
! 从这里开始，COU执行已经移动到0x9000的代码
go:	mov	ax,cs       ! 将ds、es和ss都置成移动后代码所在的段处（0x9000)。
	mov	ds,ax       ! 由于程序中有堆栈操作（push，pop，call），因此必须设置堆栈。
	mov	es,ax       
! put stack at 0x9ff00. 将堆栈指针sp指向0x9ff00（即0x9000:0xff00）处
	mov	ss,ax
	mov	sp,#0xFF00		! arbitrary value >>512
						! 由于代码移动过了，所以要重新设置堆栈的位置。
						! sp只要指向远大于512偏移（即地址0x90200）处都可以。
						! 因为从0x90200地址开始处还要放置setup程序，
						! 而此时setuo程序大约为4个扇区，因次sp要指向大于
						! (0x200 + 0x200 * 4 + 堆栈大小)处

! load the setup-sectors directly after the bootblock.
! Note that 'es' is already set up.
! 在bootsect程序块后紧跟着加载setup模块的代码数据。
! 此时es已经设置好了（移动代码时，es已经指向目的段地址处0x9000）

! 这一段的用途是利用BIOS中断的INT 0x13将setup模块从磁盘的第二个扇区
! 开始读到0x90200处，共读4个扇区。如果读出错，则复位驱动器，并重试，
! 没有退路。INT 0x13的使用方法如下：
! 读扇区：
! ah = 0x02 - 读磁盘扇区到内存
! al = 需要独出的扇区数量
! ch = 磁道（柱面）号的低8位
! cl = 开始扇区（0-5位），磁道号的高2位（6-7）
! dh = 磁头号
! dl = 驱动器号（如果是硬盘则要位7要置位）
! es:bx -> 指向数据缓冲区；如果出错则CF标志置位
load_setup:
	mov	dx,#0x0000		! drive 0, head 0
	mov	cx,#0x0002		! sector 2, track 0
	mov	bx,#0x0200		! address = 512, in INITSEG
	mov	ax,#0x0200+SETUPLEN	! service 2, nr of sectors
	int	0x13			! read it
	jnc	ok_load_setup		! ok - continue
	mov	dx,#0x0000
	mov	ax,#0x0000		! reset the diskette
	int	0x13
	j	load_setup

ok_load_setup:

! Get disk drive parameters, specifically nr of sectors/track
! 取磁盘驱动器的参数，特别是每道的扇区数量
! 取磁盘驱动器参数INT 0x13调用格式和返回信息如下：
! 返回信息：
! 如果出错则CF置位，并且ah = 状态码
! ah = 0
! al = 0
! bl = 驱动器类型（AT/PS2）
! ch = 最大磁道号的低8位
! cl = 每磁道最大扇区数（位0-5），最大磁道号高2位（位6-7）
! dh = 最大磁头数
! dl = 驱动器数量
! es:di -> 软驱磁盘参数表
	mov	dl,#0x00
	mov	ax,#0x0800		! AH=8 is get drive parameters
	int	0x13
	mov	ch,#0x00
	seg cs				! 表示下一条语句的操作数在cs段寄存器所指的段中
	mov	sectors,cx		! 保存每磁道扇区数
	mov	ax,#INITSEG
	mov	es,ax			! 因为上面取磁盘参数中断改掉了es的值，这里重新改回

! Print some inane message

	mov	ah,#0x03		! read cursor pos 读光标位置
	xor	bh,bh
	int	0x10
	
	mov	cx,#24			! 共24个字符
	mov	bx,#0x0007		! page 0, attribute 7 (normal)
	mov	bp,#msg1		! 指向要显示的字符串
	mov	ax,#0x1301		! write string, move cursor
	int	0x10			! 写字符串并移动光标

! ok, we've written the message, now
! we want to load the system (at 0x10000) 现在开始将system模块加载到0x10000（64K）处

	mov	ax,#SYSSEG
	mov	es,ax		! segment of 0x010000 es = system的段地址
	call	read_it		! 读磁盘上system模块，es位输入参数
	call	kill_motor	! 关闭驱动器马达，这样就可以知道驱动器的状态了

! After that we check which root-device to use. If the device is
! defined (!= 0), nothing is done and the given device is used.
! Otherwise, either /dev/PS0 (2,28) or /dev/at0 (2,8), depending
! on the number of sectors that the BIOS reports currently.
! 此后，我们检查要使用那个根文件系统设备（简称根设备）。如果已经制定了设备（!=0）
! 就直接使用给定的设备。否则就需要根据BIOS报告的每磁道扇区数来确定到底使用
! /dev/PS0(2,28)还是/dev/at0(2,8)

	seg cs
	mov	ax,root_dev		! 根设备号
	cmp	ax,#0
	jne	root_defined
	seg cs
	mov	bx,sectors		! 取上面保存的每磁道扇区数。
	mov	ax,#0x0208		! /dev/ps0 - 1.2Mb
	cmp	bx,#15			! 判断每磁道数是否=15
	je	root_defined	! 如果等于，则ax中就是引导驱动器的设备号
	mov	ax,#0x021c		! /dev/PS0 - 1.44Mb
	cmp	bx,#18
	je	root_defined
undef_root:				! 如果都不一样，死循环
	jmp undef_root
root_defined:
	seg cs
	mov	root_dev,ax		! 将检查过的设备号保存起来

! after that (everyting loaded), we jump to
! the setup-routine loaded directly after
! the bootblock:
! 到此，所有程序都加载完毕，跳转到被加载到bootsect后面的setup程序去

	jmpi	0,SETUPSEG	! 跳转到0x9020:0000（setup.s程序的开始处）。
						! 本程序到此结束

! 两个子程序

! This routine loads the system at address 0x10000, making sure
! no 64kB boundaries are crossed. We try to load it as fast as
! possible, loading whole tracks whenever we can.
!
! in:	es - starting address segment (normally 0x1000)
!
! 该子程序将系统模块加载到内存地址0x10000处，并确定没有跨越64KB的内存边界。
! 我们试图尽快地进行加载，只要可能，就每次记载整条磁道的数据
! 输入： es - 开始内存地址段值（通常是0x1000）

sread:	.word 1+SETUPLEN	! sectors read of current track 
							! 当前磁道中已读的扇区数。
							! 开始时已经读进1扇区的引导扇区bootsect和setup
							! 程序所占的扇区数SETUPLEN。
head:	.word 0			! current head 当前磁头号
track:	.word 0			! current track 当前磁道数

read_it:
! 测试输入的段值。从盘上读入的数据必须存放在位于内存地址64KB的边界开始处，
! 否则进入死循环。清bx寄存器，用于表示当前段内存放数据的开始位置。
	mov ax,es
	test ax,#0x0fff
die:	jne die			! es must be at 64kB boundary es值必须位于64KB地址边界
		xor bx,bx		! bx is starting address within segment bx为段内偏移位置
rp_read:
! 判断是否已经读入全部数据。比较当前所读段是否就是系统数据末端所处的段（#ENDSEG），
! 如果不是就跳转至小面ok1_read标号处继续读数据。否则退出子程序返回。
	mov ax,es
	cmp ax,#ENDSEG		! have we loaded all yet?是否已经加载了全部数据
	jb ok1_read
	ret
ok1_read:
! 计算和验证当前磁道需要读取的扇区数，放在ax寄存器中
! 根据当前磁道还未读取的扇区数以及段内数据字节开始偏移位置，计算如果
! 全部读取这些未读扇区，所读字节数是否会超过64KB段长度的限制。若会超
! 过，则根据此次最多能读入的字节数（64KB-段内偏移位置），反算出此次
! 需要读取的扇区数。
	seg cs
	mov ax,sectors		! 取每磁道扇区数
	sub ax,sread		! 减去当前磁道已读扇区数
	mov cx,ax			! cx = ax = 当前磁道未读扇区数
	shl cx,#9			! cx = cx * 512 字节
	add cx,bx			! cx = cx + 段内当前偏移值（bx）= 此次读操作后，段内共读入的字节数
	jnc ok2_read		! 若没有超过64KB字节，则跳转至ok2_read处执行
	je ok2_read
	xor ax,ax			! 若加上此次将读磁道上所有未读扇区时会超过64KB，则计算此
	sub ax,bx			! 时最多能读入的字节数（64KB - 段内读偏移位置），在转换
	shr ax,#9			! 成需要读取的扇区数
ok2_read:
	call read_track
	mov cx,ax			! cx = 该次操作已读取的扇区数
	add ax,sread		! 当前磁道上已经读取的扇区数
	seg cs
	cmp ax,sectors		! 若果当前磁道上还有扇区未读，则跳转到ok3_read处
	jne ok3_read		
	mov ax,#1			! 读该磁道的下一磁头面（1号磁头）上的数据。如果已经完成，则去读下一磁道
	sub ax,head			
	jne ok4_read		! 如果是0磁头，则再去读1磁头面上的扇区数据
	inc track			! 否则去读下一磁道
ok4_read:
	mov head,ax			! 保存当前磁头号
	xor ax,ax			! 清当前磁道已读扇区数
ok3_read:
	mov sread,ax		! 保存当前磁道已读扇区数
	shl cx,#9			! 上次已读扇区数*512字节
	add bx,cx			! 调整当前段内数据开始位置
	jnc rp_read			! 若小于64KB边界值，则跳转到rp_read处，继续读数据，否则调整当前段，为读下一段数据作准备
	mov ax,es
	add ax,#0x1000		! 将段基址调整为指向下一个64KB内存开始处
	mov es,ax
	xor bx,bx			! 清段内数据开始偏移值
	jmp rp_read			! 跳转至rp_read处，继续读数据
! 读当前磁道上指定开始扇区和需读扇区数的数据到es:bx开始处。
! al - 需读扇区数
! es:bx - 缓冲区开始位置
read_track:
	push ax
	push bx
	push cx
	push dx
	mov dx,track		! 取当前磁道号
	mov cx,sread		! 取当前磁道上已读扇区数
	inc cx				! cl = 开始读扇区
	mov ch,dl			! ch = 当前磁道号
	mov dx,head			! 取当前磁头号
	mov dh,dl			! dh = 磁头号
	mov dl,#0			! dl = 驱动器号（为0表示当前A驱动器）
	and dx,#0x0100		! 磁头号不大于1
	mov ah,#2			! ah = 2，读磁盘扇区功能号
	int 0x13
	jc bad_rt			! 若出错，则跳转至bad_rt
	pop dx
	pop cx
	pop bx
	pop ax
	ret
! 执行驱动器复位操作（磁盘中断功能号0），再跳转到read_track处重试。
bad_rt:	mov ax,#0
	mov dx,#0
	int 0x13
	pop dx
	pop cx
	pop bx
	pop ax
	jmp read_track

/*
 * This procedure turns off the floppy drive motor, so
 * that we enter the kernel in a known state, and
 * don't have to worry about it later.
 */
! 这个子程序用于关闭驱动马达，这样我们进入内核后它处于已知状态，以后也无须担心它了
kill_motor:
	push dx
	mov dx,#0x3f2		! 软驱控制卡的驱动端口，只写
	mov al,#0			! A驱动器，关闭FDC，禁止DMA和中断请求，关闭马达
	outb				! 将al中的内容输出到dx指定的端口去
	pop dx
	ret

sectors:
	.word 0				! 存放当前启动软盘每磁道的扇区数

msg1:
	.byte 13,10			! 回车、换行的ASCII码
	.ascii "Loading system ..."
	.byte 13,10,13,10	! 共24个ASCII码

.org 508				! 表示下面语句从地址508（ox1fc）开始，所以root_dev
						! 在启动扇区的第508开始的2个字节中
root_dev:
	.word ROOT_DEV		! 这里存放跟文件系统所在的设备号（init/main.c中会用）
boot_flag:
	.word 0xAA55		! 磁盘有效标识

.text
endtext:
.data
enddata:
.bss
endbss:
